
Subject: Re: Pregnancy History
Posted by [Trevor-DHS](#) on Mon, 20 Jun 2016 03:24:02 GMT
[View Forum Message](#) <> [Reply to Message](#)

S228 does exist in PKIR61FL.

See below code:

```
cd "C:\DATA\DHS_Stata"
```

```
use "PKIR61FL.DTA", clear
```

```
* Rename variables to permit reshaping
```

```
ren idx97_0* idx97_*
```

```
ren ord97_0* ord97_*
```

```
ren s2*_0* s2*_*
```

```
* keep the variables you need - you may need more than this list
```

```
keep caseid v001 v002 v003 v005 v008 v013 v024 v025 idx97_* ord97_* s2*_*
```

```
* reshape into pregnancies
```

```
reshape long idx97_ ord97_ s215_ s216_ s217_ s219_ s220m_ s220y_ ///
```

```
  s221_ s222_ s223_ s224_ s225u_ s225n_ s226m_ s226y_ s227_ s228_, i(caseid) j(idx)
```

```
* drop underscores (_) from names
```

```
ren idx97_ idx97
```

```
ren ord97_ ord97
```

```
ren s2*_ s2*
```

```
* drop empty pregnancies
```

```
drop if idx97 == .
```

```
* outcome of pregnancy
```

```
gen outcome = .
```

```
replace outcome = 1 if s216==1 | s217 == 1
```

```
replace outcome = 2 if outcome == . & s216 == 2
```

```
replace outcome = 3 if outcome == . & s228 == 1
```

```
replace outcome = 4 if outcome == . & s228 != 1
```

```
lab def outcome 1 "Live birth" 2 "Stillbirth" 3 "Abortion" 4 "Miscarriage"
```

```
lab val outcome outcome
```

```
tab outcome,m
```
